Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 18

Centers of Mass

Initializations

```
> restart;
  with(plots):
```

18.1 The continuous case

Examples

Example 18.1.1

Find the center of mass (centroid) of the region bounded by $y = 2x^2 - 3$ and $y = 3x - 4$.

Solution
First we sketch the region and find its vertices.

```
> f:=x->2*x^2-3;
```

$$f := x \to 2\,x^2 - 3 \quad \textbf{(2.1.1.1)}$$

```
> g:=x->3*x+4;
```

$$g := x \to 3\,x + 4 \quad \textbf{(2.1.1.2)}$$

```
> plot({f(x), g(x)}, x=-3..3);
```




```
> pts:=solve(f(x)=g(x), x);
```

$$pts := \frac{3}{4} + \frac{1}{4}\sqrt{65}, \frac{3}{4} - \frac{1}{4}\sqrt{65} \quad \textbf{(2.1.1.3)}$$

Since the location of the centroid is independent of the mass density ρ, we may without loss of generality assume ρ = 1. Compute the mass.

```
> Mass:=Int(g(x)-f(x), x=pts[2]..pts[1]);
  Mass:=simplify(value(Mass));
```

$$Mass := \int_{\frac{3}{4}-\frac{1}{4}\sqrt{65}}^{\frac{3}{4}+\frac{1}{4}\sqrt{65}} \left(3\,x+7-2\,x^2\right)\,dx$$

$$Mass := \frac{65}{24}\sqrt{65} \tag{2.1.1.4}$$

Compute the moment about the *y* axis.

```
> M[y]:=Int(x*(g(x)-f(x)), x=pts[2]..pts[1]);
  M[y]:=simplify(value(M[y]));
```

$$M_y := \int_{\frac{3}{4}-\frac{1}{4}\sqrt{65}}^{\frac{3}{4}+\frac{1}{4}\sqrt{65}} x\left(3\,x+7-2\,x^2\right)\,dx$$

$$M_y := \frac{65}{32}\sqrt{65} \tag{2.1.1.5}$$

Compute the moment about the *x* axis.

```
> M[x]:=Int(1/2*(g(x)^2-f(x)^2), x=pts[2]..pts[1]);
  M[x]:=simplify(value(M[x]));
```

$$M_x := \int_{\frac{3}{4}-\frac{1}{4}\sqrt{65}}^{\frac{3}{4}+\frac{1}{4}\sqrt{65}} \left(\frac{1}{2}\left(3\,x+4\right)^2-\frac{1}{2}\left(2\,x^2-3\right)^2\right)\,dx$$

$$M_x := \frac{65}{8}\sqrt{65} \tag{2.1.1.6}$$

Find the centroid.

```
> centroid:=[M[y]/Mass, M[x]/Mass];
```

$$centroid := \left[\frac{3}{4}, 3\right] \tag{2.1.1.7}$$

We can visualize the location of the centroid in the plot of *f* and *g*.

```
> p1:=plot({f(x), g(x)}, x=-3..3, color=blue):
> p2:=pointplot(centroid, style=point, symbol=box, color=
  red):
> display([p1, p2]);
```



>

18.2 The discrete case

Examples

Example 18.2.1

Find the center of gravity of the points with masses $m_1 = 5, m_2 = 7$, and $m_3 = 10$ at locations $(-1, 3)$, $(4, 8)$, and $(10, -5)$ respectively.

Solution

First we enter the *x* and *y* coordinates of the points and their masses.

```
> xp:=[-1, 4, 10];
```

$$xp := [-1, 4, 10] \quad (3.1.1.1)$$

```
> yp:=[3, 8, -5];
```

$$yp := [3, 8, -5] \quad (3.1.1.2)$$

```
> m:=[5, 7, 10];
```

$$m := [5, 7, 10] \quad (3.1.1.3)$$

Compute the mass.

```
> Mass:=Sum(m[i], i=1..3);
  Mass:=value(Mass);
```

$$Mass := \sum_{i=1}^{3} ([5, 7, 10])_i$$

$$Mass := 22 \quad (3.1.1.4)$$

Compute the moment about the *y* axis.

```
> M[y]:=Sum(xp[i]*m[i], i=1..3);
  M[y]:=value(M[y]);
```

$$M_y := \sum_{i=1}^{3} ([-1, 4, 10])_i \, ([5, 7, 10])_i$$

$$M_y := 123 \qquad (3.1.1.5)$$

Compute the moment about the *x* axis.

```
> M[x]:=Sum(yp[i]*m[i], i=1..3);
  M[x]:=value(M[x]);
```

$$M_x := \sum_{i=1}^{3} ([3, 8, -5])_i \, ([5, 7, 10])_i$$

$$M_x := 21 \qquad (3.1.1.6)$$

Compute the center of gravity.

```
> cog:=[M[y]/Mass, M[x]/Mass];
```

$$cog := \left[\frac{123}{22}, \frac{21}{22} \right] \qquad (3.1.1.7)$$

```
>
```